Derivation of DSO brightness versus distance in Celestia-FT1.2

Fridger Schrempp, Oct 30, 2005

```
> restart;
> with(plots):
```
Warning, the name changecoords has been redefined

Mostly we use the general ansatz:

$$brightness(appMag) = A - B\,appMag(absMag, d);$$

that matches the physiological eye sensitivity as to brightening of galaxies
on closer distance. d is the distance [ly] from the observer to the galaxy **border**, not to its center!
absMag is from the deepsky.dsc catalog.

First, we fix the 2 parameters A,B by requiring that

1) brightness(absMag) = r ~ 1, (with r= absMag/avgAbsMag as usual)
2) brightness(faintestMag) =1/5, (the lower visibility limit on the screen)

Solving for A,B:

```
> u:=solve({A-B*absMag=r,A-B*faintestMag=1/5},{A,B});
```
$$u := \left\{ B = \frac{1}{5}\ \frac{5\,r-1}{-absMag + faintestMag},\ A = \frac{1}{5}\ \frac{-absMag + 5\,r\,faintestMag}{-absMag + faintestMag} \right\}$$

```
> assign(u);
```

Express appMag in terms of absMag and d=distance to the galaxy's **border** (not to its center)
pc10 = 32.6167 stands for 10 pc in [ly]. d has dimensions of [ly] in Celestia.

As it should be: **appMag = absMag at d=10 pc**:

```
> appMag:=absMag  + 5 *( log10(10* d/pc10) -1);
```
$$appMag := absMag + \frac{5 \ln\left(\frac{10\,d}{pc10}\right)}{\ln(10)} - 5$$

The ansatz brightness = $A - B\,appMag$,
is assumed until a distance of **10pc** away from the galaxy **border, where appMag=absMag** by definition.

For **0< d< 10pc**, we match a brightness enhancement term, to compensate for the
"low" luminous sprite density. Then the brightness approaches a constant $r + ri$, towards the galaxy border $d = 0$;
That parameter ri also allows to adjust the brightness of the MilkyWay as seen from the solar system independently.

Take the following ansatz for the enhancement term $\frac{ri\ (pc10-d)}{pc10+d\,C}$;
Match the two functional forms by requiring their values and gradients to be equal at **d=10pc**:

```
> eq1:=subs(d=pc10,diff(ri*(pc10-d)/(pc10+d*C),d))=subs(d=pc10,diff(A-B*
  appMag,d));
```

$$eq1 := -\frac{ri}{pc10+pc10\,C} = -\frac{5\,r-1}{(-absMag+faintestMag)\ pc10\ln(10)}$$

```
> R:=solve(B=b,r);
```

$$R := \frac{1}{5} - b\,absMag + b\,faintestMag$$

solve for the parameter c in the enhancement term:

```
> c:=collect(subs(r=R,collect(solve(eq1,C),[ri,log(10)],factor)),[ri],factor);
```

$$c := -1 + \frac{1}{5}\ \frac{\ln(10)\ ri}{b}$$

ri fixes the brightness value towards the galaxy center, $r = \frac{absMag}{avgAbsMag} \sim 1$;
as usual.

Code a small procedure for plotting and calculating the resulting brightness

```
> brightness:=proc(ri,absMag,avgAbsMag,faintestMag,distance)
> local a,b,c,r,appMag;
> if type(distance,numeric) then
> r:=absMag/avgAbsMag;
> appMag:=absMag - 5 + 5 * log10(distance / 3.26167);
> a:= (1/5*absMag-r*faintestMag)/(absMag-faintestMag);
> b:= -(r-1/5)/(absMag-faintestMag);
> c:=ln(10)/5*ri/b-1;
> if (distance >= 32.6167) then
> evalf(a-b*appMag);
> else
> evalf(r + ri*(32.6167-distance)/(32.6167+distance*c));
> end if;
> else
> 'brightness(ri,absMag,avgAbsMag,faintestMag,distance)'; end if;
> end:
```

Test:

```
> brightness(0.5,-20,-16.5,16,1);
```

1.609248751

Plot for the typical values of M31

```
> semilogplot(brightness(0.5,-20.04,-19.556,7.65,distanceToDSO),distanceToDSO=
  0.01..1e7, labels=[distanceToDSO, brightness],labeldirections=[horizontal,vertical]
  );
```



Next we check the new day-night transition form, where faintestMag is decreased from ~7 at night (default) to negative values in Renderer::render().
This is at nominal distance from Earth (2.47e6 ly).

The final form used including smooth day-night blending is

brightness[eff] = brightness*(faintestMag -2)/renderer->getFaintestAM45deg(); renderer->getFaintestAM45deg() ~ 7

Since in the code **negative** brightness is set to zero, **brightness = 0 for faintestMag < 2.5** as required
for a successful, i.e natural day-night blending.

```
> plot((faintestMag-2)/7*brightness(0.5,-20.04,-19.556,faintestMag,2.47e6),
  faintestMag=-5..16,view=[-5..16,-0.5..1.5],labels=[faintestMag,brightness],
  labeldirections=[HORIZONTAL,VERTICAL]);
```


1.5
1.0
0.5
0.0
−0.5
brightness
−5
0
5
10
15
faintestMag

>